
Introduction

This document details the implementation of a framework that enable simple a way for a user to user BW queries to retrieve comments from the BW Business Document Store (BDS)

In this implementation Custom InfoObjects, a BW Cube and an ABAP class is created. The will enable you to:-

1. Use Queries to select relevant documents in the BDS.
2. Extract the Header Information of the document such as :-


· Cube in which it was created for.

· Entered Name of the comment.

· Entered Description of the comment

· KeyFigure in which it was created for.
· Query in which it was created on
· Which User created the comment.

3. Extract the comment and cut it into strings of 60 characters.

This development will allow users to query and see the comments that have been entered, this is of particular interest when commenting on aggregated data is required as this allows the user to see all comments made by other users.

Implementation

To Implement this framework you will need to either:-

Importing the development.

1. Create BW InfoObjects by loading the attached xml file in RSA1.

2. Use SAPLink to load the Nugget File - NUGG_ZCL_GET_BDS_COMMENTS.nugg (www.saplink.org ). Activate the classes ZCX_GET_BDS_COMMENTS_ERROR & ZCL_GET_BDS_COMMENTS.


3. Create the Function Module described in Appendix-C – Comments Cube Function Module


4. Create the InfoCube described in Appendix-D – Comments InfoCube
5. Activate all components

Manually.

Follow each step described in: -

Appendix-A – InfoObjects
Appendix-B – Class Implementation
Appendix-C – Comments Cube Function Module
Appendix-D – Comments InfoCube
Appendix-A – InfoObjects
This section describes how to create the custom InfoObjects needed to support this framework.

ZBDS_S001 – Document String.

Create InfoObject ZBDS_S001 Document String #1 and ensure that Lowercase Letters are enabled.

ZBDS_S002 -> ZBDS_S016 – Document String.

Create InfoObjects from ZBDS_S002 -> ZBDS_S016 (Document String #nn) as a reference characteristic of ZBDS_S001. If more characters are needed this can be extended up to ZBDS_S060.

ZBDS_NAME – BW Document Name.

Create InfoObject ZBDS_NAME – BW Document Name as a reference characteristic of ZBDS_S001.

ZBDS_DESC – BW Document Description.

Create InfoObject ZBDS_DESC – BW Document Description as a reference characteristic of ZBDS_S001.

ZBDS_CUBE – BW Document Cube.

Create InfoObject ZBDS_CUBE - BW Document Cube InfoObject.

ZBDS_GUID – BW Document GUID.

Create InfoObject ZBDS_GUID - BW Document GUID InfoObject.

ZBDS_KF – BW Document KeyFigure.

Create InfoObject ZBDS_KF - BW Document KeyFigure InfoObject.

[image: image8.png]Chararteristic

ZBD5_KF

Long deseription

EWDacument KeyFigura

Short destription

BWDacument KYFG

Version

@ [Active El

Object Status

&, wecutavis

Dictionary

Other

Dataelement  |/BIC/0IZBDS KE [Atribute Only

Data Tye CHAR - Character trin_ %1| | Persan Respans.

Lengih 30

Content release

Lowercase letters
Canvers. Rout

(m]

ALPHA ] Characteristc Is Document Praperty

Outputlength

El

siDtable [BIC/STBD5 KE Canstant





ZBDS_QRY – BW Document Query.

Create InfoObject ZBDS_QRY - BW Document Query InfoObject.

ZBDS_USER – BW Document Created User.

Create InfoObject ZBDS_USER - BW Document Created By User InfoObject.

ZBDS_RECS – BW Document Counter.

Create KeyFigure ZBDS_RECS - BW Document Created By User InfoObject.

Appendix-B – Class Implementation

ZCL_GET_BDS_COMMENTS

Use Transaction SE24 and create class ZCL_GET_BDS_COMMENTS.

Class Components

Public Section (Goto->Public Section)

*----------------------------------------------------------------------*
*       CLASS ZCL_GET_BDS_COMMENTS  DEFINITIO
*----------------------------------------------------------------------*
*
*----------------------------------------------------------------------*
CLASS zcl_get_bds_comments DEFINITION
  PUBLIC
  FINAL
  CREATE PUBLIC.

*"* public components of class ZCL_GET_BDS_COMMENTS
*"* do not include other source files here!!!
  PUBLIC SECTION.
    TYPE-POOLS rsdrc.

    DATA g_comment_cube       TYPE rsinfoprov.

    METHODS constructor
      IMPORTING
        value(i_comment_cube) TYPE rsinfoprov
      RAISING
        zcx_get_bds_comments_error.

    METHODS get_comment_data
      IMPORTING
        !ith_sfc              TYPE rsdri_th_sfc
        !ith_sfk              TYPE rsdri_th_sfk
        !it_range             TYPE rsdri_t_range
        !itx_rangetab         TYPE rsdri_tx_rangetab
        !i_first_call         TYPE rs_bool
        !i_packagesize        TYPE i
      EXPORTING
        !et_data              TYPE STANDARD TABLE
        !e_end_of_data        TYPE rs_bool
        !et_msg               TYPE rs_t_msg
      RAISING
        zcx_get_bds_comments_error.
Private Section (Goto->Private Section)

*"* private components of class ZCL_GET_BDS_COMMENTS
*"* do not include other source files here!!!
PRIVATE SECTION.

  TYPES:
    BEGIN OF _iobj_props,
           iobj               TYPE rsd_s_iobjnm,
           prop               TYPE sdok_propn,
           has_compound       TYPE rs_bool,
           p_table            TYPE rschntab,
         END OF _iobj_props.
  TYPES:
    BEGIN OF _iobj_compounds,
           p_table            TYPE rschntab,
           iobj               TYPE rsd_s_iobjnm,
           fieldnm            TYPE sdok_propn,
         END OF _iobj_compounds.

  DATA: gt_sfc                TYPE rsdri_t_sfc.
  DATA: g_comment_structure   TYPE rstlogotab.
  DATA: gt_sfk                TYPE rsdri_t_sfk.
  DATA: gt_iobj_props         TYPE TABLE OF _iobj_props.
  DATA: gt_iobj_compounds     TYPE TABLE OF _iobj_compounds.

  METHODS query_comment_data
    IMPORTING
      !it_range               TYPE rsdri_t_range
    EXPORTING
      !et_data                TYPE STANDARD TABLE
    RAISING
      zcx_get_bds_comments_error.

  METHODS _get_cube_structure
    IMPORTING
      value(i_cube)           TYPE rsinfocube
    RETURNING
      value(r_cube_structure) TYPE rsviewtiobjnm2.

  METHODS _handover_data
    IMPORTING
      !it_sfc                 TYPE rsdri_t_sfc
      !it_sfk                 TYPE rsdri_t_sfk
      !it_data                TYPE STANDARD TABLE
    EXPORTING
      !et_data                TYPE STANDARD TABLE.

  METHODS _stripout_invalid_chars
    IMPORTING
      !i_string               TYPE char4000
    RETURNING
      value(r_string)         TYPE char4000.

  METHODS _get_char_comments
    IMPORTING
      !i_string               TYPE char4000
    CHANGING
      !c_data                 TYPE any.
Update Type Descriptions

METHOD constructor.

*=============================================================================================*
*                                                                                             *
* Description                                                                                 *
* --------------------------------------------------------------------------------------------*
*                                                                                             *
* This Constructor sets the class attributes of :-                                            *
*                                                                                             *
*   1. Comments Infoprovider                                                                  *
*   2. List of InfoObjects activated for Documents                                            *
*   3. A Table of InfoObjects with Compounds and their corresponding                          *
*      Field Names                                                                            *
*                                                                                             *
*=============================================================================================*

  DATA: lt_iobj_docs   TYPE rsd_t_iobjnm,     " Document Property InfoObjects
        lt_iobj_cmp    TYPE rsd_t_iobj_cmp,   " InfoObjects Compound

        ls_cube        TYPE rsd_s_cube,       " Infoprovider Details

        ls_sfk         TYPE rsdri_s_sfk,      " InfoProviders KeyFigures
        ls_sfc         TYPE rsdri_s_sfc,      " InfoProviders Characteristics
        ls_iobj_prop   TYPE _iobj_props.

  DATA: ls_iobj_cmp    TYPE _iobj_compounds,
        l_fieldnm      TYPE rsdiobjfieldnm,
        l_iobj         TYPE rsdiobjnm.

  FIELD-SYMBOLS:
        <bi_iobjs>         TYPE rsd_s_iobjnm.

  REFRESH: gt_sfc.
*=============================================================================================*
* Set the Comments InfoProvider.
*=============================================================================================*
  g_comment_cube = i_comment_cube.

*=============================================================================================*
* Get the Comments InfoProvider Structures
*=============================================================================================*
  g_comment_structure = me->_get_cube_structure( g_comment_cube ).

*=============================================================================================*
* Get InfoObject Document Properties.
*=============================================================================================*

  lt_iobj_docs = cl_rsod_prop_generation=>get_all_docu_iobjnm( ).

  LOOP AT lt_iobj_docs ASSIGNING <bi_iobjs>.

    CLEAR: ls_iobj_prop, l_iobj, l_fieldnm. REFRESH lt_iobj_cmp.
*
* Get BDS Property Field
*
    ls_iobj_prop-iobj = <bi_iobjs>-iobjnm.
    ls_iobj_prop-prop = cl_rsod_prop_generation=>get_prop_name_from_iobjnm( i_iobjnm = <bi_iobjs>-iobjnm ).
*
* Check if InfoObject has a compound.
*
    CALL FUNCTION 'RSD_IOBJ_GET'
      EXPORTING
        i_iobjnm         = <bi_iobjs>-iobjnm
        i_objvers        = rs_c_objvers-active
      IMPORTING
        e_t_iobj_cmp     = lt_iobj_cmp
      EXCEPTIONS
        iobj_not_found   = 1
        illegal_input    = 2
        bct_comp_invalid = 3
        OTHERS           = 4.

    IF sy-subrc <> 0.
      MESSAGE ID sy-msgid TYPE sy-msgty NUMBER sy-msgno
      WITH sy-msgv1 sy-msgv2 sy-msgv3 sy-msgv4.
    ENDIF.

    IF lt_iobj_cmp IS NOT INITIAL.
*
* InfoObject does have Compounds
*
      ls_iobj_prop-has_compound = rs_c_true.
*
* Get P-Table of InfoObject
*
      CALL FUNCTION 'RSD_CHKTAB_GET_FOR_CHA_BAS'
        EXPORTING
          i_chabasnm = <bi_iobjs>-iobjnm
        IMPORTING
          e_chntab   = ls_iobj_prop-p_table
        EXCEPTIONS
          name_error = 1
          OTHERS     = 2.

      IF sy-subrc <> 0.
        MESSAGE ID sy-msgid TYPE sy-msgty NUMBER sy-msgno
        WITH sy-msgv1 sy-msgv2 sy-msgv3 sy-msgv4.
      ENDIF.
*
* Build a table of InfoObjects Key Fields for P-Table.
*
      SELECT fieldname
      INTO   l_fieldnm
      FROM   dd03l
      WHERE  tabname = ls_iobj_prop-p_table
      AND    keyflag = rs_c_true
      AND    fieldname <> 'OBJVERS'.

        CLEAR: l_iobj.

        SELECT SINGLE iobjnm
        INTO   l_iobj
        FROM   rsdiobj
        WHERE  fieldnm = l_fieldnm
        AND    objvers = 'A'.

        ls_iobj_cmp-p_table = ls_iobj_prop-p_table.
        ls_iobj_cmp-iobj    = l_iobj.
        ls_iobj_cmp-fieldnm = l_fieldnm.

        APPEND ls_iobj_cmp TO gt_iobj_compounds.

      ENDSELECT.

    ENDIF.

    APPEND ls_iobj_prop TO gt_iobj_props.

  ENDLOOP.

*=============================================================================================*
* Add Additional InfoObject Document Properties.
*=============================================================================================*

  CLEAR: ls_iobj_prop.

  ls_iobj_prop-iobj = 'ZBDS_CUBE'. ls_iobj_prop-prop = 'BW_INFOPROV'.     APPEND ls_iobj_prop TO gt_iobj_props.
  ls_iobj_prop-iobj = 'ZBDS_QRY'.  ls_iobj_prop-prop = 'BW_QUERY'.        APPEND ls_iobj_prop TO gt_iobj_props.
  ls_iobj_prop-iobj = 'ZBDS_KF'.   ls_iobj_prop-prop = 'BW_KYFNM'.        APPEND ls_iobj_prop TO gt_iobj_props.
  ls_iobj_prop-iobj = 'ZBDS_NAME'. ls_iobj_prop-prop = 'BW_DISPLAY_NAME'. APPEND ls_iobj_prop TO gt_iobj_props.
  ls_iobj_prop-iobj = 'ZBDS_DESC'. ls_iobj_prop-prop = 'DESCRIPTION'.     APPEND ls_iobj_prop TO gt_iobj_props.
  ls_iobj_prop-iobj = 'ZBDS_USER'. ls_iobj_prop-prop = 'CREATED_BY'.      APPEND ls_iobj_prop TO gt_iobj_props.

ENDMETHOD.
Method – GET_COMMENT_DATA
Parameters

Source Code

METHOD get_comment_data.

*=============================================================================================*
*                                                                                             *
* Description                                                                                 *
*---------------------------------------------------------------------------------------------*
*                                                                                             *
* This Method :-                                                                              *
*                                                                                             *
*   1. Queries the Data from the BDS                                                          *
*   2. Converts the data into OLAP Format                                                     *
*   3. Raises any exceptions if they occur.                                                   *
*                                                                                             *
*=============================================================================================*

*
* For characteristics, navigational attributes and key figures
* the non-RFC structures can be used. Instead of hashed tables
* standard tables are required.
*
  DATA: lt_sfc      TYPE rsdri_t_sfc,
        lt_sfk      TYPE rsdri_t_sfk,
        lt_range    TYPE rsdri_t_range.
*
* Variables used for Structure of InfoProvider
*
  DATA: lt_edata    TYPE REF TO data,
        ls_edata    TYPE REF TO data.
*
* Field symbols
*
  FIELD-SYMBOLS:
        <lt_edata>  TYPE STANDARD TABLE, " Infoprovider Data
        <ls_edata>  TYPE ANY.            " Infoprovider Structure

*=============================================================================================*
* Initialization of internal tables and variables
*=============================================================================================*

  lt_sfk   = ith_sfk.
  lt_sfc   = ith_sfc.
  lt_range = it_range.

*=============================================================================================*
* Determine structure of the InfoProvider
*=============================================================================================*

  CREATE DATA lt_edata TYPE TABLE OF (g_comment_structure).
  ASSIGN lt_edata->* TO <lt_edata>.

  CREATE DATA ls_edata TYPE          (g_comment_structure).
  ASSIGN ls_edata->* TO <ls_edata>.

*=============================================================================================*
* Get Comments and Populate in the structure of the Infoprovider
*=============================================================================================*

  CALL METHOD me->query_comment_data
    EXPORTING
      it_range = lt_range
    IMPORTING
      et_data  = <lt_edata>.

*=============================================================================================*
* Convert Data to OLAP Structure.
*=============================================================================================*

  CALL METHOD me->_handover_data
    EXPORTING
      it_sfc  = lt_sfc                   " Characteristics selected in query.
      it_sfk  = lt_sfk                   " Key Figures
      it_data = <lt_edata>
    IMPORTING
      et_data = et_data.

ENDMETHOD.
Method – QUERY_COMMENT_DATA
Parameters

Source Code

METHOD query_comment_data.

*=============================================================================================*
*                                                                                             *
* Description                                                                                 *
*---------------------------------------------------------------------------------------------*
*                                                                                             *
* This Method :-                                                                              *
*                                                                                             *
*   1. Queries the Data from the BDS                                                          *
*   2. Converts the data into OLAP Format                                                     *
*   3. Raises any exceptions if they occur.                                                   *
*                                                                                             *
*=============================================================================================*

*
* Variables used for Structure of InfoProvider
*
  DATA: lt_edata              TYPE REF TO data,
        ls_edata              TYPE REF TO data.
*
* Variables used for Method
*
  DATA: lr_rsod_doc           TYPE REF TO cl_rsod_doc,

        lt_query              TYPE TABLE OF sdokquprop,
        lt_result             TYPE TABLE OF sdokobject,
        lt_properties_request TYPE TABLE OF sdokproptn,
        lt_properties_result  TYPE TABLE OF sdokproptl,
        lt_iobj_docs          TYPE rsd_t_iobjnm,

        ls_query              TYPE sdokquprop,
        ls_iobj               TYPE REF TO data,
        ls_iobj_prop          TYPE _iobj_props,
        ls_io                 TYPE skwf_io,

        l_next_result         TYPE rs_bool,
        l_iobj_val            TYPE char4000,
        l_comment             TYPE char4000.
*
* Field symbols
*
  FIELD-SYMBOLS:
        <lt_edata>            TYPE STANDARD TABLE, " Infoprovider Data
        <ls_edata>            TYPE ANY.            " Infoprovider Structure

  FIELD-SYMBOLS:
        <ls_iobj>             TYPE ANY,
        <ls_iobj_comp>        TYPE ANY,
        <fieldnm>             TYPE ANY,
        <iobj>                TYPE ANY,
        <restriction>         TYPE rsdri_s_range,
        <result>              TYPE sdokobject,
        <result_properties>   TYPE sdokproptl,
        <ls_iobj_compound>    TYPE _iobj_compounds,
        <iobj_prop>           TYPE _iobj_props,
        <iobj_props>          TYPE _iobj_props.

*=============================================================================================*
* Determine structure of InfoProvider
*=============================================================================================*

  CREATE DATA lt_edata TYPE TABLE OF (g_comment_structure).
  ASSIGN lt_edata->* TO <lt_edata>.

  CREATE DATA ls_edata TYPE          (g_comment_structure).
  ASSIGN ls_edata->* TO <ls_edata>.

*=============================================================================================*
* Get Comments and Put into InfoProvider Structure
*=============================================================================================*

*
* Build BDS Query from Query Restriction.
*
  LOOP AT it_range ASSIGNING <restriction>.

    READ TABLE gt_iobj_props
      WITH KEY iobj = <restriction>-chanm
      ASSIGNING <iobj_prop>.

    IF sy-subrc = 0.
      "
      " Do not include InfoObjects with Compounds as the BDS Requires the whole Key
      "
      IF <iobj_prop>-has_compound = rs_c_false.

        MOVE-CORRESPONDING <restriction> TO ls_query.
        ls_query-prop_name = <iobj_prop>-prop.
        ls_query-option    = <restriction>-compop.
        ls_query-low       = <restriction>-low.
        APPEND ls_query TO lt_query.

      ENDIF.
    ENDIF.

  ENDLOOP.
*=============================================================================================*
* Get BW Documents
*=============================================================================================*
  CALL FUNCTION 'SDOK_LOIO_FIND_BY_QUERY'
    EXPORTING
      object_class       = 'BW_LO_TRAN'
    TABLES
      query              = lt_query
      RESULT             = lt_result
      properties_request = lt_properties_request
      properties_result  = lt_properties_result
    EXCEPTIONS
      bad_class          = 1
      bad_query          = 2
      not_authorized     = 3
      OTHERS             = 4.

  IF sy-subrc <> 0.
    MESSAGE ID sy-msgid TYPE sy-msgty NUMBER sy-msgno
    WITH sy-msgv1 sy-msgv2 sy-msgv3 sy-msgv4.
  ENDIF.
*=============================================================================================*
* Build InfoProvider Contents
*=============================================================================================*

  LOOP AT lt_result ASSIGNING <result>.
*
* Get Characteristic Values from the BI Document
*
    CLEAR <ls_edata>.
    l_next_result = rs_c_false.

    LOOP AT gt_iobj_props ASSIGNING <iobj_props>.

      READ TABLE lt_properties_result
         WITH KEY objid = <result>-objid
                  name = <iobj_props>-prop
        ASSIGNING <result_properties>.

      IF sy-subrc = 0.

        IF <iobj_props>-has_compound = rs_c_false.
*
* Assign BDS Value to InfoObject Value
*
          UNASSIGN <iobj>.
          ASSIGN COMPONENT <iobj_props>-iobj OF STRUCTURE <ls_edata> TO <iobj>. CHECK sy-subrc = 0.

          IF <iobj> IS INITIAL.
            l_iobj_val = <result_properties>-value.
            l_iobj_val = me->_stripout_invalid_chars( i_string = l_iobj_val ).
            <iobj> = l_iobj_val.
          ENDIF.
        ELSE.
*
* Assign BDS Value to InfoObject and it's compounds.
*
          CREATE DATA ls_iobj TYPE (<iobj_props>-p_table).
          ASSIGN ls_iobj->* TO <ls_iobj>.

          <ls_iobj>    = <result_properties>-value.

          LOOP AT gt_iobj_compounds ASSIGNING <ls_iobj_compound>
          WHERE   p_table = <iobj_props>-p_table
          AND     fieldnm IS NOT INITIAL.

            ASSIGN COMPONENT <ls_iobj_compound>-fieldnm OF STRUCTURE <ls_iobj>  TO <fieldnm>. CHECK sy-subrc = 0.
            ASSIGN COMPONENT <ls_iobj_compound>-iobj    OF STRUCTURE <ls_edata> TO <iobj>.    CHECK sy-subrc = 0.

            l_iobj_val = <fieldnm>.
            l_iobj_val = me->_stripout_invalid_chars( i_string = l_iobj_val ).
            <iobj> = l_iobj_val.

          ENDLOOP.

        ENDIF.
      ENDIF.

    ENDLOOP.
*
* Fill Filter Values for OLAP Query (These are not BI Document Enabled
*
    LOOP AT it_range ASSIGNING <restriction>.

      UNASSIGN <iobj>.
      ASSIGN COMPONENT <restriction>-chanm OF STRUCTURE <ls_edata> TO <iobj>. CHECK sy-subrc = 0.

      IF <iobj> IS INITIAL AND <restriction>-sign = 'I'.
        <iobj> = <restriction>-low.
      ENDIF.
*
* Try to eliminate as many invalid restrictions as possible for performance reasons.
* This will not eliminate any complex selections but the query will eliminate these.
*
      CASE <restriction>-sign.

        WHEN 'I'.

          CASE <restriction>-compop.

            WHEN 'EQ'.

              l_next_result = rs_c_true.

              LOOP AT it_range TRANSPORTING NO FIELDS
                WHERE chanm = <restriction>-chanm
                AND   sign  = 'I'
                AND   low   = <iobj>.
                l_next_result = rs_c_false.
              ENDLOOP.

            WHEN 'BT'.

              l_next_result = rs_c_true.

              LOOP AT it_range TRANSPORTING NO FIELDS
                WHERE chanm = <restriction>-chanm
                AND   sign  = 'I'
                AND   low  LE <iobj>
                AND   high GE <iobj>.
                l_next_result = rs_c_false.
              ENDLOOP.

            WHEN OTHERS.
              l_next_result = rs_c_false.

          ENDCASE.

        WHEN 'E'.

          CASE <restriction>-compop.

            WHEN 'EQ'.

              l_next_result = rs_c_false.

              LOOP AT it_range TRANSPORTING NO FIELDS
                WHERE chanm = <restriction>-chanm
                AND   sign  = 'E'
                AND   low   = <iobj>.
                l_next_result = rs_c_true.
              ENDLOOP.

            WHEN 'BT'.

              l_next_result = rs_c_false.

              LOOP AT it_range TRANSPORTING NO FIELDS
                WHERE chanm = <restriction>-chanm
                AND   sign  = 'E'
                AND   low  LE <iobj>
                AND   high GE <iobj>.
                l_next_result = rs_c_true.
              ENDLOOP.

            WHEN OTHERS.
              l_next_result = rs_c_false.

          ENDCASE.

      ENDCASE.
*
* It does not meet filter criteria exit the
*
      IF l_next_result = rs_c_true.
        EXIT.
      ENDIF.

    ENDLOOP.
*
* It does not meet filter criteria ... go to the next record.
*
    IF l_next_result = rs_c_true.
      CONTINUE.
    ENDIF.
*
* Assign Document GUID to the structure
*
    UNASSIGN <iobj>.
    ASSIGN COMPONENT 'ZBDS_GUID' OF STRUCTURE <ls_edata> TO <iobj>.

    IF <iobj> IS ASSIGNED.
      l_iobj_val = <result>-objid.
      l_iobj_val = me->_stripout_invalid_chars( i_string = l_iobj_val ).
      <iobj> = l_iobj_val.
    ENDIF.
*
* Update Keyfigure Value to the structure
*
    UNASSIGN <iobj>.
    ASSIGN COMPONENT 'ZBDS_RECS' OF STRUCTURE <ls_edata> TO <iobj>.

    IF <iobj> IS ASSIGNED.
      <iobj> = 1.
    ENDIF.
*
* Get Comment String.
*
    ls_io-class = <result>-class.
    ls_io-objid = <result>-objid.

    CREATE OBJECT lr_rsod_doc
      EXPORTING
        i_s_io = ls_io.
*
* Get Comment String Content
*
    TRY.
        CALL METHOD lr_rsod_doc->get_content
          RECEIVING
            r_content = l_comment.

        l_comment  = me->_stripout_invalid_chars( i_string = l_comment ).
        CALL METHOD me->_get_char_comments( EXPORTING i_string = l_comment
                                            CHANGING c_data    = <ls_edata>  ).
      CATCH cx_rsod.

    ENDTRY.

    APPEND <ls_edata> TO et_data.
    FREE lr_rsod_doc.

  ENDLOOP.

ENDMETHOD.
Method – _GET_CUBE_STRUCTURE

Source Code

METHOD _get_cube_structure.

*=============================================================================================*
*                                                                                             *
* Description                                                                                 *
*---------------------------------------------------------------------------------------------*
*                                                                                             *
* This Method returns the data dictionary structure of a cube                                 *
*                                                                                             *
* For Example: -                                                                              *
*                                                                                             *
* Business Content Cube - 0FIGL_C01 = /BI0/V0FIGL_C012                                        *
* Bespoke Cube          - ZFIGL_C01 = /BIC/VZFIGL_C012                                        *
* Bespoke Multiprovider - ZFIGLM01  = /BIC/VZFIGLM012                                         *
*                                                                                             *
*=============================================================================================*

  DATA: ls_cube TYPE rsd_s_cube.

  CALL FUNCTION 'RSD_CUBE_GET'
    EXPORTING
      i_infocube         = i_cube
    IMPORTING
      e_s_cube           = ls_cube
    EXCEPTIONS
      infocube_not_found = 1
      illegal_input      = 2
      OTHERS             = 4.

  IF sy-subrc <> 0.
    MESSAGE ID 'BRAIN' TYPE 'I' NUMBER 330 WITH i_cube.
    EXIT.
  ENDIF.

  r_cube_structure = ls_cube-viewtiobjnm2.

ENDMETHOD.
Method – _HANDOVER_DATA
Parameters

Source Code

METHOD _handover_data.

*=============================================================================================*
*                                                                                             *
* Description                                                                                 *
*---------------------------------------------------------------------------------------------*
*                                                                                             *
* This method converts an internal table :-                                                   *
*                                                                                             *
*       From a cube structure (it_data)                                                       *
*       To an OLAP Structure  (et_data)                                                       *
*                                                                                             *
*=============================================================================================*

  DATA: ls_sfc     TYPE rsdri_s_sfc,       " Characteristics
        ls_sfk     TYPE rsdri_s_sfk,       " Key Figures
        ls_idata   TYPE REF TO data.       " Structure of Input Table

  FIELD-SYMBOLS:
        <it_data>  TYPE ANY,               " Input Table
        <is_data>  TYPE ANY,               " Input Table Structure.

        <l_source> TYPE ANY,               " Field of Input Table Structure
        <l_target> TYPE ANY.               " Field in OLAP Table Structure
*
* Dynamically define Input Table Format
*
  CREATE DATA ls_idata LIKE LINE OF et_data.
  ASSIGN ls_idata->* TO <is_data>.
*
* For Each line of Input Table Data
*
  LOOP AT it_data ASSIGNING <it_data>.
*
* Loop on the characteristics
*
    LOOP AT it_sfc INTO ls_sfc.

      ASSIGN COMPONENT ls_sfc-chanm    OF STRUCTURE <it_data> TO <l_source>. "Characteristic Reference
      ASSIGN COMPONENT ls_sfc-chaalias OF STRUCTURE <is_data> TO <l_target>. "Characteristic Alias for OLAP

      IF sy-subrc = 0.
        <l_target> = <l_source>. "Assign Value from Characteristic structure to OLAP Structure
      ENDIF.

    ENDLOOP.
*
* Loop on the Key figures
*
    LOOP AT it_sfk INTO ls_sfk.

      ASSIGN COMPONENT ls_sfk-kyfnm    OF STRUCTURE <it_data> TO <l_source>. "Characteristic Reference
      ASSIGN COMPONENT ls_sfk-kyfalias OF STRUCTURE <is_data> TO <l_target>. "Characteristic Alias for OLAP

      IF ls_sfk-kyfalias NE '1ROWCOUNT'.

        IF sy-subrc = 0.
          <l_target> = <l_source>. "Assign Value from Characteristic structure to OLAP Structure
        ENDIF.

      ENDIF.

    ENDLOOP.

    APPEND <is_data> TO et_data. "Append Line of Input Data structure to Table with OLAP Structure.

  ENDLOOP.

ENDMETHOD.
Method – _STRIPOUT_INVALID_CHARS
Parameters

Source Code

METHOD _stripout_invalid_chars.

*=============================================================================================*
*                                                                                             *
* Description                                                                                 *
*---------------------------------------------------------------------------------------------*
*                                                                                             *
* This method deletes invalid characters from a string. This method allows Upper and Lower    *
* Case and uses the standard SAP Function for valid characters in RSKC                        *
*                                                                                             *
* If the String Contains char < then this assumes that the string is in HTML Format so will   *
* Strip all characters until a > is encountered. For Example <HEAD>ABCD<END> will return ABCD *
*                                                                                             *
*=============================================================================================*

  DATA: l_string       TYPE char4000,
        l_hstring      TYPE char4000,
        l_sep_by_space TYPE rsbool,
        l_ignore       TYPE rsbool,
        l_char         TYPE char1,
        l_char_upper   TYPE char1,
        l_str_len      TYPE i,
        l_rc           TYPE i.

*=============================================================================================*
* Loop at String and strip out invalid characters
*=============================================================================================*

  CLEAR r_string.
  l_string = i_string.
  l_str_len = STRLEN( l_string ).

  DO l_str_len TIMES.
*
* Translate Character to UPPERCASE
*
    l_char = l_string(1).
    l_char_upper = l_char.
    TRANSLATE l_char_upper TO UPPER CASE.
*
* Check if valid
*
    CALL FUNCTION 'RSKC_CHAVL_CHECK'
      EXPORTING
        i_chavl     = l_char_upper
      IMPORTING
        e_err_subrc = l_rc.

    IF l_rc = 0.
*
* If the character is a SPACE then add it on the next character concatenate
*
      IF l_sep_by_space = rs_c_true.
        CONCATENATE r_string l_char INTO r_string SEPARATED BY space.
        l_sep_by_space = rs_c_false.
      ELSE.
        CONCATENATE r_string l_char INTO r_string.
      ENDIF.

      IF l_char = ' '.
        l_sep_by_space = rs_c_true.
      ENDIF.

    ENDIF.

    SHIFT l_string BY 1 PLACES.

  ENDDO.

*=============================================================================================*
* Convert from HTML Format to Text String.
*=============================================================================================*

  IF r_string CA '<'.

    l_string  = r_string.
    l_str_len = STRLEN( l_string ).
*
* Loop Through Each character in the string.
*
    DO l_str_len TIMES.
*
* Skip to next Character if between "<" and ">".
*
      IF l_ignore = rs_c_true AND l_string(1) NE '>'.
        SHIFT l_string BY 1 PLACES.
        CONTINUE.
      ENDIF.
*
* Start of Characters to Ignore.
*
      IF l_string(1) = '<'.
        l_ignore = rs_c_true.
        SHIFT l_string BY 1 PLACES.
        CONTINUE.
      ENDIF.
*
* Possible End of Characters to Ignore
*
      IF l_string(1) = '>'.

        SHIFT l_string BY 1 PLACES.

        IF l_string(1) NE '<'.  " If the Next char is not the start of the next HTML String.

          CONCATENATE l_hstring l_string(1) INTO l_hstring SEPARATED BY space.
          l_ignore = rs_c_false.

        ENDIF.

      ELSE.

        IF l_string(1) = ''.
          l_sep_by_space = rs_c_true.
          SHIFT l_string BY 1 PLACES.
          CONTINUE.
        ENDIF.

        IF l_sep_by_space = rs_c_true.
          CONCATENATE l_hstring l_string(1) INTO l_hstring SEPARATED BY space.
          l_sep_by_space = rs_c_false.
        ELSE.
          CONCATENATE l_hstring l_string(1) INTO l_hstring.

        ENDIF.

      ENDIF.

      SHIFT l_string BY 1 PLACES.

    ENDDO.
*
* Remove bsp strings.
*
    REPLACE ALL OCCURRENCES OF '&nbsp;' IN l_hstring WITH ''.
    SHIFT l_hstring BY 1 PLACES.
    r_string = l_hstring.

  ENDIF.

ENDMETHOD.
Method – _GET_CHAR_COMMENTS
Parameters

METHOD _get_char_comments.

*=============================================================================================*
*                                                                                             *
* Description                                                                                 *
*---------------------------------------------------------------------------------------------*
*                                                                                             *
* This Method cuts the comment string into strings of upto 60 characters and puts them into   *
* the document characteristic ZBDS_S000 -> ZBDS_S070                                          *
*                                                                                             *
*=============================================================================================*

  DATA: ls_data   TYPE REF TO data,
        l_string  TYPE char4000,
        l_iobj    TYPE string,
        l_count   TYPE numc3,
        l_len     TYPE i,
        l_pos     TYPE i.

  FIELD-SYMBOLS:
        <ls_data> TYPE ANY,           " Infoprovider Structure
        <iobj>    TYPE ANY.           " InfoObject
*=============================================================================================*
* Determine structure of InfoProvider
*=============================================================================================*

  CREATE DATA ls_data TYPE (g_comment_structure).
  ASSIGN ls_data->* TO <ls_data>.
  <ls_data> = c_data.

  l_string = i_string.
  l_count = 1.

  DO 70 TIMES. " This will allow upto 70 Document characteristics
*
* Get Document String of the structure
*
    UNASSIGN <iobj>.

    CONCATENATE 'ZBDS_S' l_count INTO l_iobj.
    ASSIGN COMPONENT l_iobj OF STRUCTURE <ls_data> TO <iobj>. CHECK sy-subrc = 0.
*
* Try and reduce the splitting of words between characteristics by
* checking back until you hit a space.
*
* The minimum size of the string will be 45 characters.
*
    l_len = STRLEN( l_string ).
    l_pos = 60.

    DO 15 TIMES.

      IF l_len GE 60 AND l_string+l_pos(1) <> ''.
        l_pos = l_pos - 1.
      ELSE.
        EXIT.
      ENDIF.

    ENDDO.

    IF l_pos = 45.
      l_pos = 60.
    ENDIF.

    <iobj> = l_string(l_pos).
    CONDENSE <iobj>.

    SHIFT l_string BY l_pos PLACES.
    l_count = l_count + 1.

  ENDDO.

  c_data = <ls_data>.

ENDMETHOD. 
ZCX_GET_BDS_COMMENTS_ERROR- Comments Read Error
Create class via SE24 and activate.

Appendix-C – Comments Cube Function Module

When creating the comments InfoProvider, a function module reference was needed. 

Create this via transaction SE37. 

Copy the Function Module RS_BCT_FIGL_DATA_GET_VC1 to ZBW_GET_BDS_COMMENTS_VP, make sure that you define a custom Function Group.

 This will create a function module with the correct parameters

Function Module Parameter Settings

Update the Comment.

Import Parameters

Export Parameters

In the Functional Module Goto->Global Data

Add the following statement to the Function Module Global Data

TYPE-POOLS: rsd.

Function Module Source Code

Update the Function Module Code.
FUNCTION ZBW_GET_BDS_COMMENTS_VP .
*"----------------------------------------------------------------------
*"*"Local Interface:
*"  IMPORTING
*"     REFERENCE(I_INFOPROV) TYPE  RSINFOPROV
*"     REFERENCE(I_TH_SFC) TYPE  RSDRI_TH_SFC
*"     REFERENCE(I_TH_SFK) TYPE  RSDRI_TH_SFK
*"     REFERENCE(I_T_RANGE) TYPE  RSDRI_T_RANGE
*"     REFERENCE(I_TX_RANGETAB) TYPE  RSDRI_TX_RANGETAB
*"     REFERENCE(I_FIRST_CALL) TYPE  RS_BOOL
*"     REFERENCE(I_PACKAGESIZE) TYPE  I
*"     REFERENCE(I_BASIC_INFOPROV) TYPE  RSINFOPROV OPTIONAL
*"  EXPORTING
*"     REFERENCE(E_T_DATA) TYPE  STANDARD TABLE
*"     REFERENCE(E_END_OF_DATA) TYPE  RS_BOOL
*"     REFERENCE(E_T_MSG) TYPE  RS_T_MSG
*"----------------------------------------------------------------------

  DATA:      lr_comments_cube TYPE REF TO zcl_get_bds_comments.
  CONSTANTS: c_comments_cube  TYPE rsinfoprov VALUE 'ZCOMMENTS'.

  TRY.
*
* Build object and its attributes.
*
      CREATE OBJECT lr_comments_cube
        EXPORTING
          i_comment_cube = c_comments_cube.
*
* Read Comment data
*
      CALL METHOD lr_comments_cube->get_comment_data
        EXPORTING
          ith_sfc       = i_th_sfc
          ith_sfk       = i_th_sfk
          it_range      = i_t_range
          itx_rangetab  = i_tx_rangetab
          i_first_call  = i_first_call
          i_packagesize = i_packagesize
        IMPORTING
          et_data       = e_t_data
          e_end_of_data = e_end_of_data
          et_msg        = e_t_msg.

*======================================================================*
*                       Exception Handler                              *
*======================================================================*
    CATCH zcx_get_bds_comments_error.

      CALL FUNCTION 'RRMS_MESSAGE_HANDLING'
        EXPORTING
          i_class  = 'DBMAN'
          i_type   = 'E'
          i_number = '99'
          i_msgv1  = 'Raised Error in ZCL_BDS_GET_COMMENTS Class'
          i_msgv2  = c_comments_cube.

    CATCH cx_root.

      CALL FUNCTION 'RRMS_MESSAGE_HANDLING'
        EXPORTING
          i_class  = 'DBMAN'
          i_type   = 'E'
          i_number = '99'
          i_msgv1  = 'General Error in ZCL_BDS_GET_COMMENTS Class'
          i_msgv2  = c_comments_cube.

  ENDTRY.

  FREE lr_comments_cube.

  e_end_of_data = 'X'.

ENDFUNCTION.
Appendix-D – Comments InfoCube

In this implementation a Virtual Provider with services has been created. Create the InfoCube by using the following settings.

1. Create a InfoProvider

2. Assign the Function Module (ZBW_GET_BDS_COMMENTS_VP) that will be used.

3. Define the characteristics and KeyFigure and Time Periods.
At this point also enter the additional Document Enabled Characteristics.

4. Activate the Cube.



